
This case study looks at the re-engineering of a program to derive the date of 
Easter Sunday originally written in an early version of Cobol.

SYNOPSIS
This case study looks at the re-engineering of a program to derive the date of 

Easter Sunday originally written in an early version of Cobol.

Type: legacy re-engineering

Language: Cobol

Compiler: cobc

Skills: program re-engineering

Experience Level: novice-intermediate

April is the cruelest month.
T. S. Eliot

INTRODUCTION

The following algorithm, due to the Neapolitan astronomer Aloysius Lilius and the German 
Jesuit mathematician Christopher Clavius in the late 16th century, is used by most Western 
churches to determine the date of  Easter Sunday for any year after 1582. The first systematic 
algorithm for calculating the date of  Easter was the 'canon paschalis' due to Victorius of  
Aquitania (457 A.D.). There are many indications that the sole important application of  
arithmetic in Europe during the Middle Ages was the calculation of  Easter date, and so such 
algorithms are historically significant.

THE ALGORITHM

Algorithm E (Date of Easter)
Let Y be the year for which the date of Easter is desired.

E1. [Golden Number.] 
    G := (Y mod 19) + 1
    (G is the so-called "golden number" of the year in the 
     19-year Metonic cycle)
   
E2. [Century.] 
    C := trunc[Y/100] + 1
    (When Y is not a multiple of 100, C is the century number; 
     i.e., 1984 is in the twentieth century)

E3. [Corrections.] 
    X := trunc[3C/4] - 12
    Z := trunc[(8C+5)/25] - 5
    (X is the number of years, such as 1900, in which leap year was 
     dropped in order to keep in step with the sun. Z is a special 
     correction designed to synchronize Easter with the moon's orbit.)

E4. [Find Sunday.] 
    D := trunc(5Y/4) - X - 10.
    {March( (-D) mod 7 ) actually will be a Sunday. }

E5. [Epact.] 
    E := (11G + 20 + Z - X) mod 30. 
    If E=25, and the golden number G is greater than 11, 
       or if E=24, then increase E by 1.
    (E is the so-called "epact", which specifies when a full moon occurs.)

E6. [Find full moon.] 
    N := 44 - E. 
    If N<21 then set N := N + 30.
    (Easter is supposedly the "first Sunday following the first full 
     moon which occurs on or after March 21." Actually perturbations in 
     the moon's orbit do not make this strictly true, but we are 
     concerned here with the "calendar moon" rather than the actual moon. 
     The Nth of March is a calendar full moon.)

E7. [Advance to Sunday.] 
    N := N + 7 - ((D+N) mod 7).

E8. [Get month.] 
    If N>31, the date is (N-31) April;
    otherwise the date is N March

REF

Knuth, D., “The calculation of  Easter”, Communications of  the ACM, Vol.5(4), pp.209-210 (1962)

THE CODE

The program is coded in a older dialect of  Cobol which still fussed about fixed-formatting. 

IDENTIFICATION DIVISION.
PROGRAM-ID. DATE OF EASTER.
AUTHOR. D E KNUTH.
DATE-WRITTEN. JANUARY 22, 1962.
DATE-COMPILED. JANUARY 23, 1962.
ENVIRONMENT DIVISION.
CONFIGURATION SECTION.
SOURCE-COMPUTER. COBOLIAC.
OBJECT-COMPUTER. COBOLIAC-2, PRINTER.
SPECIAL-NAMES.
    PRINTER-OVERFLOW IS SKIP-TO-NEXT-PAGE.
INPUT-OUTPUT SECTION.
FILE-CONTROL.
    SELECT ANSWER-TABLE, ASSIGN TO PRINTER.
DATA DIVISION.
FILE SECTION.
FD ANSWER-TABLE; LABEL RECORDS ARE STANDARD; DATA
       RECORD IS EASTER-DATES.
   01  EASTER-DATES.
       02  EASTER-DAY; OCCURS 6 TIMES.
           03 MONTH; SIZE IS 5 ALPHABETIC DISPLAY CHARACTERS.
           03 FILLER; SIZE IS 1 CHARACTERS.
           03 DAYS; PICTURE IS Z9,.
           03 YEARS; PICTURE IS ZZ999.
           03 FILLER; SIZE IS 6 CHARACTERS.
WORKING-STORAGE SECTION.
77  TEMP; SIZE 6 NUMERIC COMPUTATIONAL.
77  TEMP-1; SIZE 6 NUMERIC COMPUTATIONAL.
77  BASE-YEAR; SIZE 4 NUMERIC COMPUTATIONAL.
77  LINE; SIZE 2 NUMERIC COMPUTATIONAL.
77  COLUMN; SIZE 1 NUMERIC COMPUTATIONAL.
77  COLUMN-YEAR; SIZE 4 NUMERIC COMPUTATIONAL.
77  YEAR; SIZE 4 NUMERIC COMPUTATIONAL.
77  GOLDEN-NUMBER; SIZE 2 NUMERIC COMPUTATIONAL.
77  CENTURY; SIZE 2 NUMERIC COMPUTATIONAL.
77  GREGORIAN-CORRECTION; SIZE 2 NUMERIC COMPUTATIONAL.
77  CLAVIAN-CORRECTION; SIZE 2 NUMERIC COMPUTATIONAL.
77  EXTRA-DAYS; SIZE 4 NUMERIC COMPUTATIONAL.
77  EPACT; SIZE 2 NUMERIC COMPUTATIONAL.
77  DAY; SIZE 2 NUMERIC COMPUTATIONAL.
PROCEDURE DIVISION.
CONTROL SECTION.
OUTER-LOOP.
    OPEN OUTPUT ANSWER-TABLE.
    PERFORM MIDDLE-LOOP VARYING BASE-YEAR FROM 500 BY 300
    UNTIL BASE-YEAR EQUALS 5000.
    STOP RUN.
MIDDLE-LOOP.
    PERFORM INNER-LOOP VARYING LINE FROM 0 BY 1
    UNTIL LINE EQUALS 50.
INNER-LOOP.
    PERFORM COMPUTATION VARYING COLUMN FROM 1 BY 1 UNTIL COLUMN EXCEEDS 6.
    IF LINE IS NOT EQUAL TO 49, WRITE EASTER-DATES;

    OTHERWISE WRITE EASTER-DATES BEFORE SKIP-TO-NEXT-PAGE.
COMPUTATION SECTION.
FIND-YEAR.
    MULTIPLY COLUMN BY 50 GIVING COLUMN-YEAR;
    ADD COLUMN-YEAR, BASE-YEAR, LINE, AND -50 GIVING YEAR.
FIND-GOLDEN-NUMBER.
    DIVIDE 19 INTO YEAR GIVING TEMP; 
    MULTIPLY 19 BY TEMP;
    SUBTRACT TEMP FROM YEAR GIVING GOLDEN-NUMBER THEN
    ADD 1 TO GOLDEN-NUMBER.
    IF YEAR IS LESS THAN 1583 GO TO JULIAN.
GREGORIAN.
    DIVIDE 100 INTO YEAR GIVING CENTURY; ADD 1 TO CENTURY.
    MULTIPLY CENTURY BY 3 GIVING TEMP; DIVIDE 4 INTO TEMP;
    SUBTRACT 12 FROM TEMP GIVING GREGORIAN-CORRECTION.
    SUBTRACT 18 FROM CENTURY GIVING TEMP; DIVIDE 25 INTO TEMP;
    SUBTRACT TEMP AND 16 FROM CENTURY GIVING TEMP;
    DIVIDE 3 INTO TEMP GIVING CLAVIAN-CORRECTION.
    MULTIPLY YEAR BY 5 GIVING TEMP; DIVIDE 4 INTO TEMP;
    SUBTRACT 10 AND GREGORIAN-CORRECTION FROM TEMP GIVING EXTRA-DAYS.
FUDGE-EPACT.
    MULTIPLY GOLDEN-NUMBER BY 11 GIVING TEMP;
    SUBTRACT GREGORIAN-CORRECTION FROM TEMP;
    ADD 19, CLAVIAN-CORRECTION, TEMP;
    DIVIDE 30 INTO TEMP GIVING TEMP-1;
    MULTIPLY 30 BY TEMP-1;
    SUBTRACT TEMP-1 FROM TEMP; 
    ADD TEMP, 1 GIVING EPACT.
    IF EPACT EQUALS 24 OR (25 AND GOLDEN-NUMBER IS GREATER THAN 11)
        ADD 1 TO EPACT.
    GO TO ENDING-ROUTINE.
JULIAN.
    MULTIPLY YEAR BY 5 GIVING TEMP;
    DIVIDE 4 INTO TEMP GIVING EXTRA-DAYS.
    MULTIPLY GOLDEN-NUMBER BY 11 GIVING TEMP;
    SUBTRACT 4 FROM TEMP;
    DIVIDE 30 INTO TEMP GIVING TEMP-1;
    MULTIPLY 30 BY TEMP-1;
    SUBTRACT TEMP-1 FROM TEMP;
    ADD TEMP AND 1 GIVING EPACT.
ENDING-ROUTINE.
    SUBTRACT EPACT FROM 44 GIVING DAY;
    IF DAY IS LESS THAN 21 ADD 30 TO DAY.
MAKE-DAY-SUNDAY.
    ADD DAY, EXTRA-DAYS GIVING TEMP;
    DIVIDE 7 INTO TEMP GIVING TEMP-1.
    MULTIPLY 7 BY TEMP-1;
    SUBTRACT TEMP-1 FROM TEMP;
    SUBTRACT TEMP FROM 7 GIVING TEMP;
    ADD TEMP TO DAY.
TRANSFER-ANSWER.
    IF DAY EXCEEDS 31 THEN SUBTRACT 31 FROM DAY;
        MOVE "APRIL" TO MONTH(COLUMN);
    OTHERWISE MOVE "MARCH" TO MONTH(COLUMN).
    MOVE DAY TO DAYS(COLUMN);
    MOVE YEAR TO YEARS(COLUMN).

THE ANALYSIS

The first thing is, any thought of  converting this program piece by piece the way the Fortran 
program was converted - think again. No really, it’s probably not going to work. Not all Cobol 
compilers are backwards compatible, and Knuth’s code is really old. So this needs another type of 
approach. 

What is wrong with the program and why won’t it compile? The first things of  note are that the 
program is entirely in UPPERCASE, and that the output is sent to a printer. Before anything else 
can be accomplished, these two things have to be fixed. But it’s not as easy as that (well changing 
the lowercase is). Instead of  sending the output to a printer, it should be sent to a file - but fixing 
that first seems crazy. There is also no paragraphing to improve readability. So change the 
program to lowercase, and add some paragraphing. 

It still won’t compile though. The dialect of  Cobol is too old. It goes to show that every re-
engineering problem is unique. 

A Cobol program is divided into a series of  divisions, which are themselves divided up into 
sections. There is no notion of  subroutines as found in a traditional program, more the idea of  
paragraphs. This is quite different to other programming languages. The four divisions are: 

• identification - identifies the program.
• environment - information about systems such as file I/O.
• data - information about the data/variables used in the program.
• procedure - the program statements.

Structured yes... well kind of. You can’t just declare variables anywhere, and basically all variables 
are global. Oh, and you have to create the data type for each variable. 

THE FIX

So how to re-engineer the program? First figure out what is wrong with it. Compile the original 
program and the following messages are produced:

3: Warning: author is obsolete in OpenCOBOL
4: Warning: date-written is obsolete in OpenCOBOL
5: Warning: date-compiled is obsolete in OpenCOBOL
2: Error: syntax error, unexpected date, expecting WORD or LITERAL

The first three are not an issue, the last relates to the name of  the program “Date of 
Easter”.  The offending warnings can be suppressed by removing the lines, and the program 
can be renamed Easter. Now compiling it again produces more warnings and errors. The 
program has issues with the keyword size, and the variables line, column and day. If  the re-
engineering continues along this way, it will take some time to process the entire program. Most 
of  he problems with the core part of  this program have to do with changes in Cobol syntax, and 
obsolete structures. 

Another way is to find out what each part of  the program does, and re-engineer it piece-by-piece. 
Start with a blank template, containing each of  the most relevant sections:

identification division.
program-id. easter.

environment division.
    
input-output section.
file-control.
    
data division.
file section.
           
working-storage section.

procedure division.

THE environment 

Firstly the environment division. Instead of  outputting to a printer, the program is modified 
to output to file. Therefore the original code in file-control becomes:

select efile assign to "easter.out"
    organization is line sequential.

This code assigns the variable efile to the external file easter.out.

THE data 

Next the data division. How should the data be formatted for output? This is dealt with in the 
file section. The old code looks like this:

fd answer-table; label records are standard; data
        record is easter-dates.
    01  easter-dates.
        02  easter-day; occurs 6 times.
            03 month; size is 5 alphabetic display characters.
            03 filler; size is 1 characters.
            03 days; picture is z9,.
            03 years; picture is zz999.
            03 filler; size is 6 characters.

Part of  the problem lies in obsolete code (shown in blue), and methods for declaring the type of  
variables. The keyword fd is synonymous with a file descriptor. This is modified to:

M.Wirth © 2014



fd efile.
01  easter-record.
    03  easter-month    pic x(5).
    03  a-space         pic x.
    03  easter-day      pic z9,.
    03  easter-year     pic zz999.

The old code deals with this in terms of  a structure containing six of  these records, the new code 
contains a single record. This essentially creates a record named easter-record, and 
associates it with the file descriptor efile. This record actually produces output of  the form:

! March 31, 2013

in the file easter.out. The first element of  the record prints 5 alphabetic characters. This is 
followed by a space, the day of  Easter (a number between 1 and 31) followed by a comma, and 
the year of  Easter left-padded with a space. Basically it looks like:

! xxxxxxz9,zz999

The first z9 pads allows both 1-digit (padded with a space to the left) and 2-digit numbers to be 
printed. The zz999 allows the same with years 500 to 5000, left padding unused spots with 
spaces.

Next the variables are dealt with. The variables in the old  working-storage section 
look like this:

77  temp; size 6 numeric computational.
77  temp-1; size 6 numeric computational.
77  base-year; size 4 numeric computational.
77  line; size 2 numeric computational.
77  column; size 1 numeric computational.
77  column-year; size 4 numeric computational.
77  year; size 4 numeric computational.
77  golden-number; size 2 numeric computational.
77  century; size 2 numeric computational.
77  gregorian-correction; size 2 numeric computational.
77  clavian-correction; size 2 numeric computational.
77  extra-days; size 4 numeric computational.
77  epact; size 2 numeric computational.
77  day; size 2 numeric computational.

The variables line, column and day are all Cobol keywords and will need to be modified 
(although the first two will disappear shortly, so they can be left as is). They can be modified by 
appending -1 to the variable name - so day becomes day-1. (Note that use of  the - in variable 
names is not unusual in Cobol). The re-engineered code looks like this:

77 temp                  pic 9(6).
77 temp-1                pic 9(6).
77 base-year             pic 9(4).

77 line                  pic 99.
77 column                pic 9.
77 column-year           pic 9(4).
77 year                  pic 9(4).
77 golden-number         pic 99.
77 century               pic 99.
77 gregorian-correction  pic 99.
77 clavian-correction    pic 99.
77 extra-days            pic 9(4).
77 epact                 pic 99.
77 day-1                 pic 99.

At this stage, no spurious variables have been removed, it is a one-to-one conversion. 

THE procedure 

The original code uses a convoluted piece of  code to process each of  the years from 500 to 5000 
- basically three loop type paragraphs, not untypical of  Cobol. It looks like this:

outer-loop.
    open output answer-table.
    perform middle-loop varying base-year from 500 by 300
    until base-year equals 5000.
    stop run.
    
middle-loop.
    perform inner-loop varying line from 0 by 1
    until line equals 50.
    
inner-loop.
    perform computation varying column from 1 by 1 until column exceeds 6.
    if line is not equal to 49, write easter-dates;
    otherwise write easter-dates before skip-to-next-page.

The paragraph outer-loop opens the output stream, and performs the paragraph middle-
loop 15 times, the paragraph middle-loop performs the paragraph inner-loop 50 times,  
and the paragraph inner-loop calls the paragraph computation 6 times for a total of  
4500 iterations.  Outputting the dates of  Easter Sunday in a linear fashion it much simpler. Here 
is how the code changes:

    open output efile.
    perform computation 
        varying year from 500 by 1
        until year is equal to 5000. 
    close efile.
    stop run.

This basically iterates the value of  the variable year from 500 to 5000, calculating the date of  
Easter Sunday for each year.  Modifying this code to simplify it has a number of  advantages:

1. As the variables base-year, line, column are part of  the old code, they are no longer 
needed.

2. In the computation section, the paragraph named find-year, which uses all these 
variables, and the variable column-year to calculate the actual year is no longer needed (and 
neither is the variable column-year).

77 temp                  pic 9(6).
77 temp-1                pic 9(6).
77 base-year             pic 9(4).
77 line                  pic 99.
77 column                pic 9.
77 column-year           pic 9(4).
77 year                  pic 9(4).
77 golden-number         pic 99.
77 century               pic 99.
77 gregorian-correction  pic 99.
77 clavian-correction    pic 99.
77 extra-days            pic 9(4).
77 epact                 pic 99.
77 day-1                 pic 99.

find-year.
    multiply column by 50 giving column-year;
    add column-year, base-year, line, and -50 giving year.

The code so far is shown in the program e1.cob.

THE computation section 

Now that the core data and file I/O has been dealt with, the actual computational paragraphs 
can be re-engineered. In the old code, the program flowed sequentially in the following manner 
in the computation section:

find-year.
find-golden-number.
    > if year < 1583 go to julian.
gregorian.
fudge-epact.
    > go to ending-routine.
julian.
ending-routine.
make-day-sunday.
transfer-answer.

In the re-engineered code, the paragraph computation runs the core paragraphs in the program:

computation.
    perform find-golden-num.
    perform ending-routine.

    perform make-day-sunday.
    perform transfer-answer.
    write easter-record.

This suppresses the two go to statements.The modules julian, gregorian and fudge-
epact are called from find-golden-num. Now each paragraph can be re-engineered.

find-golden-number.
    divide 19 into year giving temp; 
    multiply 19 by temp;
    subtract temp from year giving golden-number then
    add 1 to golden-number.
    if year is less than 1583 go to julian.

Re-engineering the code involves modifying the syntax of  the calculations. In this case the 
statements are turned into compute statements. This means a statement like:

    divide 19 into year giving temp; 

becomes

    compute temp = year / 19.

It makes the Cobol program seem less “wordy” - and easier to interpret. Care should be taken 
when converting such calculations.

find-golden-num.
    compute temp = year / 19.
    compute temp = temp * 19.
    compute golden-number = year - temp.
    add 1 to golden-number.

    if year1 is < 1583 then
        perform julian
    else
        perform gregorian
        perform fudge-epact
    end-if.

In the original code, the if  statement passes control to the paragraph julian if  the value of  
year is less than 1583. If  not, control just passed to the next sequential paragraph, gregorian. 
In the modified code, an if-else statement passes control to the appropriate paragraph. Note 
that failure to rewrite the if-else statement will result in a warning of  the form:

! Warning: IF statement not terminated by END-IF

Below is the re-engineering of  the paragraph julian:

julian.
    multiply year by 5 giving temp;
    divide 4 into temp giving extra-days.
    multiply golden-number by 11 giving temp;
    subtract 4 from temp;
    divide 30 into temp giving temp-1;
    multiply 30 by temp-1;
    subtract temp-1 from temp;
    add temp and 1 giving epact.

julian.
    compute temp = year * 5.
    compute extra-days = temp / 4.
    compute temp = golden-number * 11.
    subtract 4 from temp.
    compute temp-1 = temp / 30.
    compute temp-1 = temp-1 * 30.
    subtract temp-1 from temp.
    compute epact = temp + 1.

Below is the re-engineering of  the paragraph gregorian:

gregorian.
    divide 100 into year giving century; add 1 to century.
    multiply century by 3 giving temp; divide 4 into temp;
    subtract 12 from temp giving gregorian-correction.
    subtract 18 from century giving temp; divide 25 into temp;
    subtract temp and 16 from century giving temp;
    divide 3 into temp giving clavian-correction.
    multiply year by 5 giving temp; divide 4 into temp;
    subtract 10 and gregorian-correction from temp giving extra-days.

gregorian.
    compute century = year / 100.
    add 1 to century.
    compute temp = century * 3.
    divide 4 into temp.
    compute gregorian-correction = temp - 12.
    compute temp = century - 18.
    divide 25 into temp.
    compute temp = century - temp - 16.
    compute clavian-correction = temp / 3.
    compute temp = year * 5.
    divide 4 into temp.
    compute extra-days = temp - 10 - gregorian-correction.

Below is the re-engineering of  the paragraph fudge-epact:

fudge-epact.
    multiply golden-number by 11 giving temp;
    subtract gregorian-correction from temp;
    add 19, clavian-correction, temp;
    divide 30 into temp giving temp-1;
    multiply 30 by temp-1;
    subtract temp-1 from temp; 
    add temp, 1 giving epact.
    if epact equals 24 or (25 and golden-number is greater than 11)
        add 1 to epact.
    go to ending-routine.

fudge-epact.
    compute temp = golden-number * 11.
    subtract gregorian-correction from temp.
    compute temp = temp + 19 + clavian-correction.
    compute temp-1 = temp / 30.
    multiply 30 by temp-1.
    subtract temp-1 from temp.
    compute epact = temp + 1.
    if epact is = 24 or (epact is = 25 and golden-number is > 11) then
        add 1 to epact
    end-if.

Below is the re-engineering of  the paragraph ending-routine:

ending-routine.
    subtract epact from 44 giving day-1;
    if day-1 is less than 21 add 30 to day-1.

ending-routine.
    subtract epact from 44 giving day-1.
    if day-1 is < 21 then
        add 30 to day-1
    end-if.

Below is the re-engineering of  the paragraph make-day-sunday:

make-day-sunday.
    add day-1, extra-days giving temp;
    divide 7 into temp giving temp-1.
    multiply 7 by temp-1;
    subtract temp-1 from temp;
    subtract temp from 7 giving temp;
    add temp to day-1.

make-day-sunday.
    compute temp = day-1 + extra-days.
    compute temp-1 = temp / 7.
    multiply 7 by temp-1.
    subtract temp-1 from temp.
    compute temp = 7 - temp.
    add temp to day-1.

Below is the re-engineering of  the paragraph transfer-answer:

transfer-answer.
    if day-1 exceeds 31 then subtract 31 from day-1;
        move "april" to month(column);
    otherwise move "march" to month(column).
    move day-1 to days(column);
    move year to years(column).

transfer-answer.
    if day-1 is > 31 then
        subtract 31 from day-1
        move "April" to easter-month
    else
        move "March" to easter-month
    end-if.
    move day-1 to easter-day.
    move year to easter-year.

The code so far is shown in the program e2.cob.

MAKING CODE COMPACT

The question is, can the statements be modified in such a manner - maybe compression - that 
would make the code more compact (and maybe more readable) ? Consider the portion of  the 
algorithm which calculates the Gregorian correction (inside paragraph Gregorian).

! X := trunc[3C/4] - 12

This is represented in the re-engineered code as:

! compute temp = century * 3.
! divide 4 into temp.
! compute gregorian-correction = temp - 12.

However three lines of  code to represent one line in an algorithm almost seems like overkill.  If  
this can be represented as:

! compute temp = century * 3.
! computer temp = temp / 4.
! compute gregorian-correction = temp - 12.

Then all three lines can be compressed into:

! compute gregorian-correction = ((century * 3) / 4) - 12.

The same can be done for the Clavian correction:

! compute temp = century - 18.
! divide 25 into temp.
! compute temp = century - temp - 16.
! compute clavian-correction = temp / 3.

Then all four lines can be compressed into:

! compute clavian-correction = (century -((century - 18) / 25) - 16) / 3.

And similarly for extra-days. A paragraph with 12 lines of  code originally, now only has four.

gregorian.
    compute century = (year / 100) + 1.
    compute gregorian-correction = ((century * 3) / 4) - 12.
    compute clavian-correction = (century -((century - 18) / 25) - 16) / 3.
    compute extra-days = ((year * 5) / 4) - 10 - gregorian-correction.

This can be done similarly throughout the program.

NOTE: If  you use a dash in a variable, the compiler will not accept something like:

! temp-temp-1

Obviously, because it thinks it’s a variable. It has to be written as:

! temp - temp-1

Other code, such as legacy code to calculate a modulus such as:

! compute temp = year / 19.
! compute temp = temp * 19.
! compute golden-number = year - temp.
! add 1 to golden-number.

Can be replaced with code which uses the function mod:

! compute golden-number = function mod(year,19) + 1.

Note that code such as:

! compute golden-number = (year - ((year / 19) * 19)) + 1.

Does not work (returns a value of  1) - and is ridiculous anyway. Note also that code such as:

! compute clavian-correction = (century - ((century-18) / 25) - 16) / 3.

does not work in the same way as:

! compute temp = (century - 18) / 25.
! compute clavian-correction = (century - temp - 16) / 3.

If  century = 50, then the first statement produces a result of  10. The calculated result is 10.9, 
which is truncated to 10 when assigned to clavian-correction.This is because the value 

The next two statements produce a result of  11, because the first statement calculates a value of  
1.28 which is truncated to 1 when assigned to temp. It all hinges on the type of  data created in 
working-storage. For example, consider the following code:

! compute temp = golden-number * 11.
! subtract gregorian-correction from temp.
! compute temp = temp + 19 + clavian-correction.

The variable temp is a numeric of  size 6, or 999999. So if  the following values are used:

! golden-number=02, gregorian-correction=25, clavian-correction=11

then calculations proceed as follows:

! temp = 2 * 11 = 22
! temp = temp - gregorian-correction = 22 - 25 = -3

BUT, because temp can only hold positive values, the value assigned to temp is 3. This makes all 
the difference in the next calculation:

! temp = 3 + 19 + 11 = 33

IF for instance all three statements are combined:

! compute temp = ((golden-number * 11) - gregorian-correction) 
                     + 19 + clavian-correction.
  
The result is 27, because the middle portion:

! (golden-number * 11) - gregorian-correction 

evaluates to -3, because it is within the RHS of  the statement. The three lines of  code can be 
compacted to:

! compute temp = (golden-number * 11) - gregorian-correction
! compute temp = temp + 19 + clavian-correction.

The code so far is shown in the program e3.cob.

IMPROVING READABILITY

The final thing to do is to modify some of  the variable names to shorten them. Particularly 

! golden-number, gregorian-correction, clavian-correction

Change these to:

! goldenNum, gregorianCor, clavianCor

Some of  the more confusing variable names can also be modified:

! extra-days ! extraDays
! day-1 ! easterDay

The code so far is shown in the program e4.cob.

THE FINAL CODE
identification division.
program-id. easter.

environment division.
    
input-output section.
file-control.
select efile assign to "easter.out"
    organization is line sequential.
    
data division.
file section.
fd efile.
01  easter-record.
    03  easter-month    pic x(5).
    03  a-space         pic x.
    03  easter-day      pic z9,.
    03  easter-year     pic zz999.
           
working-storage section.
77 temp              pic 9(6).
77 temp-1            pic 9(6).
77 year              pic 9(4).
77 goldenNum         pic 9(2).
77 century           pic 9(2).
77 gregorianCor      pic 9(2).
77 clavianCor        pic 9(2).
77 extraDays         pic 9(4).
77 epact             pic 9(2).
77 easterDay         pic 9(2).

procedure division.
    open output efile.
    perform computation 
        varying year from 500 by 1
        until year is equal to 5000. 
    close efile.
    stop run.
    
computation.
    perform find-golden-num.
    perform ending-routine.
    perform make-day-sunday.
    perform transfer-answer.
    write easter-record.

find-golden-num.
    compute goldenNum = function mod(year,19) + 1.

    if year is < 1583 then
        perform julian

    else
        perform gregorian
        perform fudge-epact
    end-if.
    
julian.
    compute extraDays = (year * 5) / 4.
    compute temp = (goldenNum * 11) - 4.
    compute temp-1 = temp - function mod(temp, 30).
    compute epact = (temp - temp-1) + 1.
    
gregorian.
    compute century = (year / 100) + 1.
    compute gregorianCor = ((century * 3) / 4) - 12.
    compute temp = (century - 18) / 25.
    compute clavianCor = (century - temp - 16) / 3.
    compute extraDays = ((year * 5) / 4) - 10 - gregorianCor. 
    
fudge-epact.
    compute temp = (goldenNum * 11) - gregorianCor
    compute temp = temp + 19 + clavianCor.
   
    compute epact = function mod(temp, 30) + 1.
  
    if epact is = 24 or (epact is = 25 and goldenNum is > 11) then
        add 1 to epact
    end-if.
    
ending-routine.
    compute easterDay = 44 - epact.
    if easterDay is < 21 then
        add 30 to easterDay
    end-if.
    
make-day-sunday.
    compute temp =  7 - function mod((easterDay + extraDays), 7).
    add temp to easterDay.

    
transfer-answer.
    if easterDay is > 31 then
        subtract 31 from easterDay
        move "April" to easter-month
    else
        move "March" to easter-month
    end-if.
    move easterDay to easter-day.
    move year to easter-year.
    move spaces to a-space.

